
Введение

Для того чтобы многократно использовать

модели блоков в различных частях проекта

и затем использовать ранее написанные мо-

дели блоков в новых проектах, необходимо

выполнять модели блоков параметризируе-

мыми. Такие стандартные параметры моду-

ля, как размер — SIZE, разрядность (шири-

на) — WIDTH и объем (глубина) — DEPTH,

обычно выполняются параметризируемыми

и могут переопределяться в зависимости от

того места в проекте пользователя, где они оп-

ределены. В описании параметризированных

моделей на языке Verilog используют две кон-

струкции — глобальное макроопределение

(`define) и утверждение для переопределения

параметра (defparam). В этом разделе будут

рассмотрены новые расширения языка

Verilog-2001, которые позволяют более удоб-

но выполнять кодирование, а также избежать

ошибок при выполнении кодирования.

В Verilog-1995 [см. 2-й пункт списка лите-

ратуры из части 2], есть два способа опреде-

лить константы: параметр (parameter) — это

константа, которая является локальной по от-

ношению к модулю, и макроопределение, со-

зданное путем использования директивы

компилятора, — `define.

После того, как параметр объявлен, на него

можно ссылаться, используя имя параметра.

Так же и на макроопределение `define мож-

но ссылаться после того, как оно определено,

используя название макрокоманды с предыду-

щим символом ̀  (back-tic) апострофа. Различие

в проекте между параметром и макроопределе-

нием в том, что макроопределение имеет ̀ iden-

tifier_name, в то время как параметр — только

identifier_name, без символа апострофа.

Параметры

Параметры должны быть определены в пре-

делах границ модуля, при использовании клю-

чевого слова — parameter. Повторим, что па-

раметр — это константа, которая применяет-

ся локально в данном модуле и может быть

произвольно переопределена при установке

данного компонента (instance) в проект поль-

зователя. Для описаний параметризованных

модулей, выполненных в стиле Verilog-1995,

одно или более объявлений параметров

обычно предшествуют объявлениям порта,

например, так, как показано в модуле регис-

тра (пример 1).

module register (q, d, clk, rst_n);

parameter SIZE=8;

output [SIZE-1:0] q;

input [SIZE-1:0] d; 

input clk, rst_n;

reg [SIZE-1:0] q;

always @(posedge clk or negedge rst_n)

if (!rst_n) q <= 0;

else q <= d;

endmodule

Пример 1. Параметризованная модель регистра,

выполненная в стиле Verilog!1995

В варианте исполнения этого же модуля

в стиле Verilog-2001 список параметров мо-

жет быть вынесен в заголовок модуля так же,

как это делается в стиле ANSI-C (пример 2).

module register2001 #(parameter SIZE=8)

(output reg [SIZE-1:0] q,

input [SIZE-1:0] d,

input clk, rst_n);

always @(posedge clk, negedge rst_n)

if (!rst_n) q <= 0;

else q <= d;

endmodule

Пример 2. Параметризированная модель регистра

(стиль Verilog!2001)

Параметры 
и переопределение параметра

Применяя модули с параметрами, выпол-

ненными в стандарте Verilog-1995, у пользо-

вателя есть два способа изменить параметры

как для отдельных модулей, так и для всех

примененных в проекте пользователя ком-

понентов для данных модулей.

Первый — это переопределение параметра

непосредственно в инстанции компонента,

второй — отдельным определением defparam.

В версии Verilog-2001 добавлен третий, более

мощный метод для того, чтобы изменить па-

раметры на установленных компонентах мо-

дулей: при использовании именованных пара-

метров в непосредственно установленных ком-

понентах модулей (см. часть 7 данной статьи).

Использование символа # 
при переопределении параметра

Для переопределения параметра в установ-

ленном компоненте модуля используется

символ #. Он необходим для того, чтобы ука-

зать, что параметры данного модуля должны

быть переопределены.

В примере 3 две копии регистра из приме-

ра 1 установлены в модуле two_regs1. Пара-

метр SIZE для обоих случаев устанавливает-

ся равным 16, то есть при переопределении

параметра назначением # (16) параметр оста-

ется тем же самым.

module two_regs1 (q, d, clk, rst_n);

output [15:0] q;

input [15:0] d;

input clk, rst_n;

wire [15:0] dx;

register #(16) r1(.q(q), .d(dx), .clk(clk), .rst_n(rst_n));

register #(16) r2(.q(dx), .d(d), .clk(clk), .rst_n(rst_n));

endmodule

Пример 3. Переопределение параметра

при использовании установленного компонента модуля

Эта форма переопределения параметра под-

держивалась всеми инструментальными сред-

ствами синтеза много лет. Самая большая

проблема с этим типом переопределения па-

раметра состоит в том, что параметры нужно

передать в установленный компонент моду-

ля в том же порядке, в котором они появля-

ются в установленном компоненте модуля.

Рассмотрим модуль myreg (пример 4).

module myreg (q, d, clk, rst_n);

parameter Trst = 1,

Tckq = 1,

SIZE = 4, 

VERSION = «1.1»;

output [SIZE-1:0] q;

input [SIZE-1:0] d;

input clk, rst_n;

reg [SIZE-1:0] q;

always @(posedge clk or negedge rst_n)

if (!rst_n) q <= #Trst 0;

else q <= #Tckq d;

endmodule

Пример 4. Модуль с четырьмя параметрами

В этом разделе более подробно рассмотрено создание параметризируе(

мых моделей.

Краткий курс HDL. 

Часть 9. 

Параметризируемые модели

Модуль myreg из примера 4 имеет четыре

параметра, и если при установке компонен-

та модуля требуется изменить только третий

параметр (например, SIZE), модуль не может

быть установлен так, чтобы вместо тех пара-

метров, которые не меняются, писались бы

только запятые, а для тех параметров, кото-

рые мы хотим поменять, вводились бы но-

вые значения. Например, если мы хотим по-

менять только один параметр SIZE так, как

показано в примере 5, то это было бы син-

таксической ошибкой. 

module bad_wrapper (q, d, clk, rst_n);
output [7:0] q;
input [7:0] d;
input clk, rst_n;
// illegal parameter passing example
myreg #(,,8) r1 (.q(q), .d(d), .clk(clk), .rst_n(rst_n));

endmodule

Пример 5. Переопределение параметра таким образом:

# (,,8) — это синтаксическая ошибка

Чтобы правильно использовать синтаксис

переопределения параметра при установке

нового компонента модуля, в нем должны

быть перечислены все значения параметров,

включая и те значения, которые должны быть

изменены. Для модуля myreg (пример 4)

должны быть перечислены первые два зна-

чения параметра, даже при том, что они не

изменяются. Новое значение для параметра

SIZE должно быть задано так, как показано

в примере 6.

module good_wrapper (q, d, clk, rst_n);
output [7:0] q;
input [7:0] d;
input clk, rst_n;
// the first two parameters must be
// explicitly passed even though the
// values did not change

myreg #(1,1,8) r1 (.q(q), .d(d), .clk(clk), .rst_n(rst_n));
endmodule

Пример 6. Правильный синтаксис 

при переопределении параметров: # (1,1,8)

Зная об этих ограничениях, инженеры ча-

сто перестраивали порядок следования пара-

метров в модуле таким образом, чтобы наи-

более часто используемые параметры были

помещены в модуле первыми.

Несмотря на ограничения в Verilog-1995 по

переопределению параметров, этот метод все

же лучше всего поддерживается в компиля-

торах и симуляторах. И он наболее прост для

того, чтобы изменить параметры установлен-

ного компонента модуля.

Verilog-2001 фактически расширяет выше-

упомянутые возможности по переопределе-

нию параметров, добавляя способность пе-

редать параметры по имени, подобно тому,

как производят подключения портов по их

имени.

Скажем «прощай» для defparams!

У многих пользователей языка Verilog пер-

вые впечатления, получаемые от применения

определений defparam, очень благоприятны.

Утверждение defparam явно идентифициру-

ет установленный компонент модуля и каж-

дый его индивидуальный параметр. При этом

каждый индивидуальный параметр модуля

должен быть переопределен с каждым ут-

верждением defparam. Утверждение defparam

может быть помещено перед установленным

компонентом модуля, после установленного

компонента модуля или где-нибудь еще

в файле.

До 2000 года инструментальные средства

фирмы Synopsys не разрешали использо-

вать утверждения defparam для переопре-

деления параметров. И разработчикам, ис-

пользующим инструментальные средства

от Synopsys, необходимо было помнить об

этих ограничениях. Но сейчас в последних

версиях инструментальных средств Synopsys

уже разрешено использовать утверждения

defparam.

Однако утверждение defparam часто мо-

жет быть неправильно употреблено в следу-

ющих случаях:

1. Использование defparam для того, чтобы

в иерархическом порядке изменить пара-

метры модуля.

2. Размещение определения defparam в от-

дельном файле для изменяемого установ-

ленного компонента модуля.

3. Использование многократных определе-

ний defparam в одном и том же файле, что-

бы изменить параметры установленного

компонента модуля.

4. Использование многократных утвержде-

ний defparam в нескольких различных

файлах, чтобы изменять параметры уста-

новленного компонента модуля.

Иерархические определения 
defparam

Правильный способ выполнения проек-

тов — это использование определения def-

param для того, чтобы в иерархическом по-

рядке изменить значения параметров. Это

означает, что любой параметр в проекте мо-

жет быть изменен из любого файла, входя-

щего в данный проект. Потенциально, не-

правильная запись может вызывать изме-

нение значения параметра модуля, если

в установленном компоненте модуля есть

определение defparam и определение пара-

метра — parameter, который, в свою оче-

редь, повторно изменяет параметр в уста-

новленном компоненте модуля.

В примере 7 на примере модуля testbench

(tb_defparam) показано то, как изменяется

параметр SIZE для модуля регистра (в соот-

ветствии с параметром WIDTH), который пе-

редается как параметр WIDTH для модуля dff

(передается к параметру N). Модуль dff име-

ет ошибочное иерархическое определение

defparam, которое изменяет параметр SIZE

от 8 до 1 в testbench, и это значение переда-

ется иерархически, чтобы снова изменить

в регистре значение WIDTH и значения dffN.

module tb_defparam;
parameter SIZE=8;
wire [SIZE-1:0] q;
reg [SIZE-1:0] d;
reg clk, rst_n;
register2 #(SIZE) r1(.q(q), .d(d), .clk(clk),

.rst_n(rst_n));
// ...

endmodule

module register2 (q, d, clk, rst_n);
parameter WIDTH=8;
output [WIDTH-1:0] q;
input [WIDTH-1:0] d;
input clk, rst_n;

dff #(WIDTH) d1(.q(q), .d(d), .clk(clk), .rst_n(rst_n));
endmodule

module dff (q, d, clk, rst_n); parameter N=1;
output [N-1:0] q;

input [N-1:0] d;
input clk, rst_n;
reg [N-1:0] q;

// dangerous, hierarchical defparam
defparam tb_defparam.SIZE = 1;

always @(posedge clk or negedge rst_n)
if (!rst_n) q <= 0;
else q <= d;

endmodule

Пример 7. Опасное использование иерархического

определения defparam

Все порты и переменные в проекте из приме-

ра 7 теперь имеют разрядность шины только

один бит, в то время как при синтезе модулей

register2 и dff их разрядность будет равна вось-

ми битам. Этот тип использования defparam

вызывает проблемы при проектировании и от-

ладке, и он может легко быть обнаружен.

module register3 (q, d, clk, rst_n);

parameter WIDTH=8;

output [WIDTH-1:0] q;

input [WIDTH-1:0] d;

input clk, rst_n;

dff3 #(WIDTH) d1

(.q(q), .d(d), .clk(clk),

.rst_n(rst_n));

endmodule

module dff3 (q, d, clk, rst_n);

parameter N=1;

output [N-1:0] q;

input [N-1:0] d;

input clk, rst_n;

reg [N-1:0] q;

// dangerous, hierarchical defparam

defparam register3.WIDTH = 1;

always @(posedge clk or negedge rst_n)

if (!rst_n) q <= 0;

else q <= d;

endmodule

Пример 8. Опасное иерархическое использование

определение defparam, включенное в модели, — 

register3/dff3

Пример 8 подобен примеру 7 за исключе-

нием того, что определение defparam пере-

определяет разрядность шин в модели regis-

ter3 и замыкается само на себя. К сожалению,

даже притом, что эта модель при моделиро-

вании имеет разрядность в 1 бит, при синте-

зе она выполняется как модель, имеющая раз-

рядность в 8 бит.

www.kit(e.ru

Использование определения 
defparam в отдельных файлах

Весьма обычны случаи неправильного при-

менения определения defparams, когда эти

определения помещаются в совершенно дру-

гие файлы, а не в тот файл, где находится ус-

тановленный компонент модуля, который

и должен быть изменен.

К сожалению, эта практика была вызвана

следующим комментарием в разделе 12.2.1

документов стандарта Verilog-1995 [см. 2-й

пункт списка литературы из части 2]

и Verilog-2001 [см. 1-й пункт списка литера-

туры из части 2]: «Определения defparam осо-

бенно полезно сгруппировать вместе, так,

чтобы они могли переопределить все пара-

метры в модуле».

Но на самом деле есть превосходная воз-

можность передачи параметров по имени для

установленных компонентов модулей, по-

добно тому, как это делается с определением

портов по имени. Таким образом, при ис-

пользовании передачи параметров по имени,

определения defparam в конечном счете пе-

рестанут использоваться.

Многократное использование 
определений defparam 
в одном и том же файле

В предыдущем разделе мы рассмотрели ва-

риант многократного использования def-

param в разных файлах. Теперь рассмотрим,

что произойдет при использовании много-

кратных определений defparam в одном и том

же файле. Если несколько определений def-

param помещено в одном и том же файле,

и при этом все они изменяют один и тот же

параметр, то такое применение определения

defparam тоже будет неправильно. Стандарт

Verilog-2001 определяет правильное действие

следующим образом: «В случае многократно-

го использования определения defparam для

единственного параметра, параметр берет зна-

чение последнего определения defparam, с ко-

торым сталкивается в исходном тексте».

В примере 9 две копии регистра (register),

приведенного в примере 1, установлены в мо-

дуле two_regs2. Параметр SIZE в обоих слу-

чаях установлен равным 16 при использова-

нии определения defparam, которое помеще-

но перед соответствующими установленными

компонентами регистра (register). Третье оп-

ределение defparam располагают после вто-

рого установленного компонента регистра,

что изменяет разрядность второго регистра.

Теперь разрядность второго регистра стала

равна 4, а это ошибка.

Поскольку приведенный пример представ-

ляет собой очень маленький проект, и ком-

пиляторы выдадут сообщение об ошибке

(«Размер порта не соответствует»), то этот

проект нетрудно будет отладить.

К сожалению, часто в больших проектах,

где объем RTL-кода измеряется многими

страницами, второе «беспризорное» опреде-

ление defparam может быть добавлено по

ошибке, и оно переопределит значение пара-

метра. Это произойдет только потому, что

проектировщик не обращал внимания на то,

что использовалось более раннее определе-

ние defparam. Такой способ выполнения про-

ектов является более сложным и более труд-

ным в отладке.

Многократное применение 
определений defparam 
в отдельных файлах

Иногда определения defparam могут быть

неправильно многократно размещены в раз-

личных файлах.

Большие проблемы вызывает практика

размещения многократных определений def-

param в различных файлах, которые задают

значения на один и тот же параметр. Много-

кратные определения defparam могут быть

обработаны по-разному различными изго-

товителями программных инструментов, по-

тому что алгоритм обработки для этого сце-

нария никогда не определялся в стандарте

Verilog-1995.

Группа разработчиков стандарта Verilog-2001

не хотела поощрять такое поведение, поэто-

му к стандарту Verilog-2001 была добавлена

следующая оговорка: «Когда определения def-

param находятся во многих исходных фай-

лах, например, они могут быть найдены при

поиске библиотечных файлов, то определе-

ние defparam, по которому параметр берет

свое значение, будет неопределенно».

Группа разработчиков стандарта Verilog-2001

хотела пресечь такую практику в целом, по-

этому они оставили назначение параметров

неопределенными и документально зареги-

стрировали этот факт, надеясь отговорить

любого производителя программных инст-

рументов от поддержки этой некорректной

стратегии. 

Определения defparam 
и инструментальные средства

Как уже было показано, определение def-

param может быть помещено в любом мес-

те проекта, потому что оно может в иерар-

хическом порядке изменить значения пара-

метров любого модуля в проекте. Поэтому

применение определений defparam создает

очень большие трудности при создании про-

граммного инструмента, как для продажи

пользователям в качестве товарного продук-

та, так и при применении фирмой для соб-

ственных нужд. Эти программные инстру-

менты должны очень точно анализировать

проект, в котором разрешено включить оп-

ределения defparam.

Компилятор Verilog не может определить

фактические значения любых параметров до

тех пор, пока все входные файлы проекта не

прочитаны компилятором, потому что чте-

ние последнего файла может изменить каж-

дый отдельный параметр в дизайне.

Поэтому, чтобы уменьшить вероятность

появления ошибок в проектах из-за непра-

вильного порядка обработки файлов проек-

та программными инструментами, можно

сформулировать следующее правило: «Не ис-

пользуйте определения defparam в любых

проектах, выполненных на языке Verilog».

Альтернатива определениям defparam при-

ведена в следующем разделе статьи.

Переопределение параметра 
по имени в стандарте Verilog(2001

К стандарту Verilog-2001 было добавлено

следующее расширение: это возможность ус-

тановки компонентов модулей с параметра-

ми, причем назначение параметра произво-

дится по имени параметра, непосредственно

в самом установленном компоненте модуля.

И теперь это расширение совершенно устра-

няет потребность в определениях defparam.

module demuxreg (q, d, ce, clk, rst_n);

output [15:0] q;

input [ 7:0] d;

input ce, clk, rst_n;

wire [15:0] q;

wire [ 7:0] n1;

not u0 (ce_n, ce);

regblk #(.SIZE( 8)) u1

(.q(n1), .d (d), .ce(ce), .clk(clk), .rst_n(rst_n));

regblk #(.SIZE(16)) u2

(.q (q), .d({d,n1}), .ce(ce_n), .clk(clk), .rst_n(rst_n));

endmodule

module regblk (q, d, ce, clk, rst_n);

parameter SIZE = 4;

output [SIZE-1:0] q;

input [SIZE-1:0] d;

input ce, clk, rst_n;

reg [SIZE-1:0] q;

always @(posedge clk or negedge rst_n)

if (!rst_n) q <= 0;

else if (ce) q <= d;

endmodule

Пример 10. Пример использования передачи

параметров по имени

Преимущество новой методики состоит

в том, что она предлагает указывать, какой

именно параметр будет изменен (подобно то-

му, как это делает определение defparam)

и так же, как Verilog-1995, помещает значе-

ния переопределенного параметра # в син-

таксис установленного компонента модуля.

module two_regs2 (q, d, clk, rst_n);

parameter SIZE = 16;

output [SIZE-1:0] q;

input [SIZE-1:0] d;

input clk, rst_n;

wire [SIZE-1:0] dx;

defparam r1.SIZE=16;

register r1 (.q(q), .d(dx), .clk(clk), .rst_n(rst_n));

defparam r2.SIZE=16;

register r2 (.q(dx), .d(d), .clk(clk), .rst_n(rst_n));

defparam r2.SIZE=4; // Design error!

endmodule

Пример 9. Пример использования определения 

defparam

Поскольку вся информация параметра

включена в установленный компонент моду-

ля (инстанцию модели), то такой способ за-

дания параметров абсолютно «понятен» для

любого программного инструмента, и это

именно та методика, которая должна быть

использована при разработке моделей мно-

гократного использования.

Правило: требуйте, чтобы все прохождение

параметров в вашем проекте было выполнено

в соответствии с новыми методиками пере-

определения параметра при использовании на-

званий параметров в стиле Verilog-2001. 

Макроподстановка `define (Macro)

Директива компилятора `define применяет-

ся для того, чтобы выполнить «глобальную»

макроподстановку, подобную директиве #de-

fine в языке Cи. Макроподстановки глобаль-

ны, начиная от пункта определения, и остают-

ся активными для всего этапа чтения и обра-

ботки файлов. Макроподстановки начинаются

с того момента, как только это макроопреде-

ление сделано, и действуют до тех пор, пока

другое макроопределение не изменит значе-

ние определенной макрокоманды или же по-

ка макрокоманда не будет отменена исполь-

зованием директивы компилятора `undef.

Макроопределения могут существовать как

внутри, так и снаружи объявления любого

модуля, и оба эти варианта будут обработа-

ны одинаково. Объявления же параметра мо-

гут быть сделаны только в границах модуля.

Начиная с того момента, когда макроопре-

деления будут определены и в течение всего

чтения файлов, после того, как макроопреде-

ление было определено, использование мак-

роопределений вообще делает компилирова-

ние зависимым от порядка расположений

файлов дизайна при компиляции.

Типичная проблема, связанная с использо-

ванием макроопределений, состоит в том, что

несколько файлов могут иметь макроопреде-

ления с теми же самыми названиями макро-

команды. Когда это происходит, компилято-

ры Verilog выдают предупреждения, связанные

с «переопределением макрокоманды», но, ес-

ли разработчик не заметит такое предупреж-

дение, это может привести к большим пробле-

мам при выполнении проекта или отладке.

Почему же плохо переопределить макро-

определение? Язык Verilog позволяет иметь

иерархические ссылки идентификаторов. Это

оказывается очень ценным для исследования

и отладки проекта. Если тому же самому на-

званию макрокоманды дать повторное опре-

деление в проекте, то только последнее опре-

деление будет доступно в testbench для ис-

следования и отладки проекта.

Если вы делаете многократные макроопре-

деления на одно и то же название макроко-

манды, то вы наверняка полагаете, что мак-

рокоманда должна трактоваться как локаль-

ный параметр (parameter), а не как глобальная

макрокоманда.

Использование `define

Как уже было сказано, компиляторы язы-

ка Verilog должны сначала прочитать все ма-

кроопределения в проекте и только после это-

го производить обработку остальных фай-

лов проекта. Соответственно, чтение всех

макроопределений в первую очередь при

сборке проекта обеспечивает то, что макро-

определения будут существовать всегда, ког-

да они необходимы, и что они являются гло-

бально доступными для всех файлов, собран-

ных в проекте.

Обратите внимание на предупреждения

о переопределении макрокоманды. Для ус-

пешного использования макроопределений

можно рекомендовать следующие правила:
•• Используйте макроопределения только для

идентификаторов, которые однозначно

требуют глобального определения иденти-

фикатора, который, в свою очередь, не бу-

дет изменен в другом месте проекта.
•• Где только возможно, поместите все макро-

определения в один файл “definitions.vh” и,

собирая проект, читайте этот файл первым.

Примечание: поместите все макроопределе-

ния в модуль testbench верхнего уровня и, со-

бирая проект, читайте этот файл первым.
•• Не используйте макроопределения для то-

го, чтобы определить локальные для дан-

ного модуля константы.

Включение `define

Перед тем, как использовать макроопреде-

ление, обычно выполняется проверка того

факта, что само это макроопределение суще-

ствует. И для этого есть общепринятая мето-

дика: в проекте необходимо использовать ди-

рективу компилятора `ifdef или в новом сти-

ле Verilog-2001 — `ifndef. Она делает запрос

о существовании макроопределения, сопро-

вождаемого или назначением макрокоман-

ды по `define, или `include по имени файла,

который и содержит требующееся макро-

определение (пример 11).

`ifdef CYCLE

// do nothing (better to use `ifndef)

`else

`define CYCLE 100

`endif

`ifndef CYCLE

`include «definitions.vh»

`endif

Пример 11. Проверка и отладка макроопределений

Директива компилятора `undef

Для того чтобы удалить макроопределе-

ние, созданное с помощью директивы ком-

пилятора `define, Verilog имеет директиву

компилятора `undef.

Если есть вероятность того, что сделанное

в данном файле макроопределение может

быть переопределено в другом файле, и, что-

бы избежать такой опасной ситуации, нуж-

но сделать следующее. Можно поместить оп-

ределение `undef в конце того файла, где про-

изведено определение макрокоманды по

директиве `define. Таким образом, использо-

вание пары `define – `undef следует считать

самым лучшим средством для решения дан-

ной проблемы.

Определение цикла синхрочастоты

Для того чтобы определить периоды син-

хрочастоты, можно применить либо задание

параметров (parameter), либо использовать

директиву компилятора `define. Однако ис-

пользование `define — предпочтительнее.

Правило: используйте директиву компиля-

тора `define для того, чтобы задать опреде-

ление тактового сигнала (пример 12).

Правило: поместите определения такто-

вого сигнала в файл “definitions.vh” или в test-

bench верхнего уровня (пример 13).

`define CYCLE 10

Пример 12. Определение тактового сигнала

`define CYCLE 10

module tb_cycle;

// ...

initial begin

clk = 1'b0;

forever #(`CYCLE/2) clk = ~clk;

end

// ...

Endmodule

Пример 13. Глобальное макроопределение тактового

сигнала и его рекомендованное описание

Определение длительности цикла синхро-

частоты как макроопределения имеет пре-

имущество в том, что синхрочастота — это

фундаментальная константа проекта и тест-

бенча. Цикл общего синхронизирующего сиг-

нала не должен изменяться от одного моду-

ля к другому. Длительность цикла должна

быть постоянной!

Профессиональные пользователи Verilog де-

лают большинство симуляций и проверок

в тестбенче именно на фронтах синхроимпуль-

са. Поэтому такой тип тестбенча легко масшта-

бируется в соответствии с изменениями в гло-

бальном определении тактового сигнала.

Не применяйте `define 
в автоматах состояний (FSM)

В некоторых литературных источниках

приводятся рекомендации по использованию

директивы компилятора `define для того, что-

бы определить названия состояний в описа-

нии конечного автомата на языке Verilog.

Но при разработке конечного автомата ис-

пользование параметров предпочтительнее.

Для того чтобы определить названия со-

стояний автомата (FSM), целесообразно ис-

пользовать параметры, именно потому, что

названия состояний автомата — это своего

рода константы, которые применяются толь-

ко в модуле, содержащем данный автомат.

Если в большом проекте имеется много ко-

нечных автоматов, то обычно хочется мно-

гократно использовать определенные назва-

ния состояний в нескольких описаниях авто-

мата, например RESET, IDLE, READY, READ,

WRITE, ERROR и DONE.

Использование директивы `define для то-

го, чтобы назначить название состояния,

препятствовало бы многократному его ис-

пользованию. Возможно, в глобальном про-

странстве имен требуемое вам название уже

было взято. Правда, в таком случае можно

было бы сделать следующее: применить

между модулями к названиям состояния ди-

рективу `undef и произвести повторное име-

нование состояний по директиве `define

в новых модулях автомата. Но тогда мы по-

лучим множество автоматов с одинаковы-

ми названиями состояний, а это мешает ис-

следовать внутренние назначения шин со-

стояния автоматов в тестбенче и выполнять

сравнение диаграмм различных автоматов

с одинаковыми названиями состояний. Вот

почему нет убедительной причины для оп-

ределения названий состояний по директи-

ве `define. И, следовательно, названия состо-

яний нельзя считать частью глобального

пространства имени. Названия состояний

нужно счесть местными названиями в том

модуле автомата, который их содержит.

Правило: не используйте директиву мак-

роопределения `define для назначений назва-

ний состояний автомата. 

Правило: для назначений символических

названий состояний автомата используйте

параметры — parameter.

Локальные параметры (localparam)
в стиле Verilog(2001

К стандарту Verilog-2001 было добавлено но-

вое расширение, локальный параметр — lo-

calparam. В отличие от параметров (parame-

ter), локальный параметр (localparam) не мо-

жет быть изменен переопределением параметра

(выполненное в установленном компоненте

как переопределение по позиции или по назва-

нию), и при этом localparam не может быть пе-

реопределен определением defparam. 

Локальный параметр (localparam) может

быть определен в терминах параметров, ко-

торые могут быть изменены переопределе-

нием параметра по позиции в установленном

компоненте, переопределением параметра по

имени (что является более предпочтитель-

ным) или определениями defparam.

Идея применения localparam в том, что это

позволит изменять некоторые местные зна-

чения параметров, основанных на других па-

раметрах, защищая localparam от случайно-

го или неправильного переопределения ко-

нечным пользователем.

В примере14 показано, что размер масси-

ва памяти mem должен быть сгенерирован

в соответствии с разрядностью адресной ши-

ны. Разрядность параметра объема памяти

MEM_DEPTH будет «защищена» от неправиль-

ной настройки параметров в том случае, если

объявление параметра MEM_DEPTH будет по-

мещено в localparam. Параметр MEM_DEPTH

станет иным только при изменении парамет-

ра ASIZE.

module ram1 #(parameter ASIZE=10,

DSIZE=8)

(inout [DSIZE-1:0] data,

input [ASIZE-1:0] addr,

input en, rw_n);

// Memory depth equals 2**(ASIZE)
localparam MEM_DEPTH = 1<<ASIZE;
reg [DSIZE-1:0] mem [0:MEM_DEPTH-1];

assign data = (rw_n && en) ? mem[addr]
: {DSIZE{1'bz}};

always @(addr, data, rw_n, en)
if (!rw_n && en) mem[addr] = data;

endmodule

Пример 14. Использование локального параметра 

(localparam) в соответствии со стандартом Verilog!2001.

Локальный параметр MEM_DEPTH защищен,

он вычисляется в зависимости от значения 

параметра разрядности адресной шины

Необходимо отметить, что стандарт

Verilog-2001 не распространяет возможности

расширения localparam на список парамет-

ров в заголовке модуля. Поэтому в настоя-

щее время localparam не может быть добав-

лен к списку параметров в стиле ANSI, как

это показано в примере 15.

module multiplier2

#(parameter AWIDTH=8, BWIDTH=8,

localparam YWIDTH=AWIDTH+BWIDTH)

(output [YWIDTH-1:0] y,

input [AWIDTH-1:0] a,

input [BWIDTH-1:0] b);

assign y = a * b;

endmodule

Пример 15. Неправомерное использование параметра

localparam в заголовке модуля в стиле ANSI

Определение шкалы времени —
`timescale

Директива шкалы времени `timescale дает

значение задержкам, которые могут появить-

ся в Verilog-модели . Шкала времени помеща-

ется выше заголовка модуля и имеет следу-

ющий вид (пример 16).

`timescale time_unit / time_precision

Пример 16. Директива шкалы времени

Директива шкалы времени (`timescale) име-

ет огромное влияние на производительность

большинства симуляторов языка Verilog. Час-

то встречающаяся ошибка у новичков при ис-

пользовании симуляторов состоит в том, что

они выбирают time_precision 1ps (1 пикосе-

кунда) для того, чтобы тщательно рассмотреть

на диаграммах сигналов в своем проекте, из-

менения за каждую пикосекунду времени мо-

делирования. Применяя при моделировании

точность в 1ps, 1ns или 100ps time_precisions,

время моделирования может увеличиться бо-

лее чем на 100%, и при этом использование па-

мяти увеличится более чем на 150%.

Довольно часто для того, чтобы облегчить

изменение всей шкалы времени в проекте, раз-

работчики используют макроопределение. Все

модули в таком проекте включают в себя ма-

крокоманду шкалы времени перед каждым за-

головком модуля. В примере 17 показано ма-

кроопределение для глобальной шкалы вре-

мени (`timescale) и использование глобальной

макрокоманды шкалы времени (`timescale).

`define tscale `timescale 1ns/1ns

`tscale

module mymodule (...);

...

Пример 17. Глобальное определение макрокоманды

шкалы времени (не рекомендуется к применению)

Пользователи обычно надеются управлять

эффективностью моделирования, изменяя

глобальное определение шкалы времени

(`timescale), чтобы одновременно иметь воз-

можность изменить и time_units, и time_pre-

cisions для каждой модели, входящей в про-

ект, и, таким образом, повысить производи-

тельность симулятора.

Однако необходимо иметь в виду, что гло-

бальное изменение time_units каждой шка-

лы времени (`timescale) во всем проекте мо-

жет неблагоприятно повлиять на его целост-

ность. Любой проект, который включает

#delays, полагается на точность указанного

time_units в директиве шкалы времени.

В примере 18 требуется, чтобы интервалы

времени time_units шкалы времени были

равны 100ps. Изменение time_units до значе-

ния в 1ns изменяет задержку от 160ps до 1.6ns,

что приводит к ошибке в модели.

`timescale 100ps/10ps

module tribuf2001 #(parameter SIZE=8)

(output [SIZE-1:0] y,

input [SIZE-1:0] a,

input en_n);

assign #1.6 y = en_n ? {SIZE{1'bz}}:a;
endmodule

Пример 18. Модуль с time_units по 100ps

Так как величина time_precision всегда

должна быть равна или меньше, чем time_unit

в директиве шкалы времени, то при исполь-

зовании глобальной стратегии по назначению

шкалы времени необходимо придерживаться

следующих дополнительных рекомендаций.

Правило: сделайте все определяемые поль-

зователем time_units для шкалы времени рав-

ными или большими, чем 1ns.

Причина здесь в том, что, если в какой-ни-

будь модели будет использоваться меньшее

значение time_unit, то, глобально изменяя все

значения time_precision до величины в 1ns,

можно нарушить весь существующий проект.

Необходимо отметить следующее: если

модель, поставляемая другими фирмами,

будет включена в проект, и если в этой мо-

дели использовалось очень маленькое зна-

чение time_precision, то все моделирование

сильно замедлится и его производительность

будет крайне низкая, даже в том случае, ес-

ли в пользовательских моделях значение

time_precision будет глобально изменено.

Заключение

Макроопределения должны использовать-

ся для того, чтобы определить глобальные кон-

станты для всей системы, так, чтобы пользо-

вателю было удобно и привычно ими пользо-

ваться, например, названия для команд PCI или

глобальных определений тактового цикла.

Каждый раз, как только будет сделано но-

вое макроопределение, название этой макро-

команды уже нельзя безопасно использовать

в другом месте в проекте (загрязнение про-

странства имени). Поскольку в большие про-

екты для моделирования может быть собра-

но все больше и больше модулей, то, соот-

ветственно, увеличивается и вероятность

«столкновения» макроимен. Практика ис-

пользования макроопределений для констант

типа порта или размеров данных и названий

состояний может привести к ошибке.

Макроопределения с директивой компи-

лятора `define не должны использоваться для

определения констант, которые гораздо удоб-

нее могут быть определены в модулях инди-

видуально.

Параметры localparam в Verilog предназна-

чены для того, чтобы представить константы,

которые являются для модуля локальными.

Локальные параметры имеют дополнитель-

ные преимущества в том, что каждый отдель-

ный установленный компонент модуля может

иметь различные значения для параметров.

Приведем перечень наиболее важных ре-

комендаций, приведенных в этом разделе:
•• Во всех проектах, выполненных на языке

Verilog, не используйте defparams.
•• Требуйте, чтобы все определение параме-

тров в проекте было выполнено при ис-

пользовании методики переопределения

параметров по имени, в соответствии с но-

вым стандартом Verilog-2001.
•• Используйте макроопределения для иден-

тификаторов, которые требуют только гло-

бального определения идентификатора, та-

кого, который не будет изменен в другом

месте в проекте.
•• Где только возможно, поместите все макро-

определения в один файл “definitions.vh” и,

собирая проект, читайте этот файл первым.

Примечание: поместите все макроопределе-

ния в модуль testbench верхнего уровня и, со-

бирая проект, читайте этот файл первым.
•• Не используйте макроопределения для то-

го, чтобы определить константы, которые

являются местными в конкретном модуле.
•• Сделайте определения тактового цикла, ис-

пользуя директиву компилятора `define.
•• Поместите определения тактовой частоты

в файл “definitions.vh” или в testbench верх-

него уровня. 
•• Не используйте назначения для названий

состояний автомата при макроопределе-

ниях `define.
•• Назначайте состояния, используя параме-

тры parameter с символическими названи-

ями состояний.
•• Чтобы улучшать эффективность модели-

рования, сделайте все определяемые поль-

зователем time_units для шкалы времени

равными или большими, чем 1ns.

В следующем разделе мы рассмотрим вопро-

сы, связанные с обработкой сигнала «сброс». ■

